

ニュートン方程式

Copyright ©2010 by Shigeto R. Nishitani

ニュートンはティコ・ブラーヘやケプラーの太陽系の観測を基に、粒子、剛体球の古典的な運動方程式を導いた。太陽系の運動は直感的に理解しやすいので、それを記述するケプラーの法則から見ていく。

ケプラーの法則の視覚化


ケプラーの法則は

第1法則（楕円軌道の法則）：惑星は、太陽をひとつの焦点とする楕円軌道上を動く。

第2法則（面積速度一定の法則）：惑星と太陽とを結ぶ線分が単位時間に描く面積は、一定である。

第3法則（調和の法則）：惑星の公転周期の2乗は、軌道の長半径の3乗に比例する。である。このうち、第1と第2法則を視覚化すると下記の通りとなる。

```
> restart;
tmp:=[[1.995499248, -0.8992494792e-1], [1.978599987,
-.1892845822], [1.949096870, -.2874362623], [1.906844688,
-.3837042244], [1.851630204, -.4773585802], [1.783167578,
-.5675901158], [1.701092302, -.6534784844], [1.604953846,
-.7339503458], [1.494207473, -.8077222107], [1.368206607,
-.8732196335], [1.226199090, -.9284591568], [1.067335289,
-.9708698302], [.8907075786, -.9970133803], [.6954697665,
-1.002129121], [.4811638197, -.9793665893], [.2486093915,
-.9184625248], [0.2412795050e-2, -.8035400533],
[-.2417377506, -.6105284679], [-.4382395350,
-.3123373982], [-.4967632174, 0.7424236313e-1],
[-.3743717192, .4363904666], [-.1504738434, .6926486454],
[0.9860825922e-1, .8522751524], [.3414384128,
.9436229580], [.5679019134, .9876637201], [.7753432123,
.9979336736], [.9636343432, .9832536630], [1.133475444,
.9495924413], [1.285785717, .9011552434], [1.421480224,
.8410175569], [1.541390963, .7715052948], [1.646243831,
.6944315164], [1.736657303, .6112494185], [1.813148875,
.5231547540], [1.876143624, .4311567338], [1.925982479,
.3361288119], [1.962929432, .2388463368], [1.987177233,
.1400155883], [1.998851625, 0.4029719115e-1]]];
> with(plots):with(plottools):
potential:=plot3d(-1/sqrt(x^2+y^2)-0.1,x=-1..2.5,y=-1.5..
1.5,view=-2.5..0,style=wireframe);
potential:=PLOT3D(...) (1.1.1)
> planet3:=proc(x)
global tmp,e;
plots[display]([potential,
sphere([tmp[x][1],tmp[x][2],-1/sqrt(tmp[x][1]^2+tmp[x][
2]^2)],0.1)], style=patchnograd);
end;
> tmp2:=[];
for i from 1 to 39 do
tmp2:=op(tmp2),planet3(i);
end do;
tmp2:=[] (1.1.2)
> display(tmp2,insequence=true,scaling=constrained,
orientation=[-90,0], axes=none);
```



この図を少し傾けると重力ポテンシャルの様子が示されている。楕円軌道や面積速度が一定となる原因が直観的に理解できるのでは？

分子動力学法の基本

惑星の軌道は摩擦のない宇宙空間での2体の重力だけで考えることができる単純な動きなので、比較的容易に再現することができる。この運動を記述する方程式がニュートンの運動方程式

$$F=ma$$

である。ここで F はForce（力）、 m はmass（質量）、 a はacceleration（加速度）である。 a を微分で明示すると、

$$F=m\frac{d^2}{dt^2}r(t)$$

となる。 $r(t)$ がある時間での位置をさしているとする。これを微小時間 h で級数展開して $t+h$ と $t-h$ での和をとると、 $t+h$ での位置を $r(t-h)$ と $r(t)$ での情報から予測することができる。

```
> eq1:=r(t+h)=subs(x=t+h,series(r(x),x=t,3));
eq1:=r(t+h)=r(t)+D(r)(t)h+1/2D(2)(r)(t)h^2+O(h^3) (1.2.1)
```

```
> eq2:=r(t-h)=subs(x=t-h,series(r(x),x=t,3));
eq2:=r(t-h)=r(t)-D(r)(t)h+1/2D(2)(r)(t)(-h)^2+O((-h)^3) (1.2.2)
```

```
> eq3:=convert(eq1+eq2,polynomial);
eq3:=r(t+h)+r(t-h)=2r(t)+D(2)(r)(t)h^2 (1.2.3)
```

```
> eq4:=solve(eq3,r(t+h));
eq4:=2r(t)+D(2)(r)(t)h^2-r(t-h) (1.2.4)
```

```
> r(t+h)=sort(subs((D@2)(r)(t)=F/m,eq4),h,ascending);
```

$$r(t+h) = 2r(t) - r(t-h) + \frac{Fh^2}{m} \quad (1.2.5)$$

このように微分方程式を差分方程式に変換し、数値的に解いていく方法がVerlet法であり、そのほかの種々の分子動力学法の手法の基本となる考え方である。

Verlet法による解

実際の粒子（剛体球）の運動を記述するには粒子にかかる F （力）を知る必要がある。惑星の軌道を考えるときには単純に太陽を中心とする重力

$$F = -G \frac{mM}{d^2}$$

であるので非常に単純になる。 G は重力定数、 m, M は2体それぞれの質量、 d は2体間の距離である。

適当に規格化して、惑星の位置を $r=[x,y]$ とすると、距離 $d=\sqrt{x^2+y^2}$ として、力は

$$F = 1/d^2$$

で与えられる。その力ベクトルを位置によって分解すると、 $[-x/d, -y/d]$ となるので、

$$L = (x^2+y^2)^{3/2};$$

$$\text{force} = [-x/L, -y/L];$$


で求めることができる。

初期値を $r(0-h), r(0)$ に適当におけば、後は(1.2.5)式に従って芽ずる式で惑星の未来の軌跡をシミュレーションすることができる。

```
> restart;
with(plots):
> force:=proc(pos)
  local x,y,L;
  x:=pos[1];
  y:=pos[2];
  L:=(x*x+y*y)^(3/2);
  return [-x/L,-y/L];
end proc;
> Verlet:=proc(r0,rh)
  global m,h;
  local f,x,y;
  f:=force(r0);
  x:=2*r0[1]-rh[1]+h^2/m*f[1];
  y:=2*r0[2]-rh[2]+h^2/m*f[2];
  return [x,y];
end proc;
> dx:=0.0;dy:=0.01;
h:=0.01;
m:=0.2;
r:=[[2,0],[2-dx,-dy]];
for i from 1 to 392 do
  r:=[op(r),Verlet(r[-1],r[-2])];
end do;
dx:=0;
dy:=0.01;
h:=0.01;
m:=0.2;
r:=[[2,0],[2,-0.01]]
```

(1.3.1)

```
> pointplot(r,scaling=constrained);
```



```
> r[-1];
```

[2.000016724, 0.0003039509108] (1.3.2)

```
> nn:=nops(r);
tmp1:=[];
dens:=10;
for i from 1 to nn/dens do
  tmp1:=[op(tmp1),pointplot(r[i*dens])];
end;
nn:=394;
tmp1:=[];
dens:=10
```

(1.3.3)

```
> nops(tmp1);
```

39 (1.3.4)

```
> display(tmp1,insequence=true,scaling=constrained);
```